

Beispiel 2 Lineal Modalanalyse, Transiente

Lernziel

In diesem Beispiel kann der FEM-Neuling ein Experiment und eine Simulation am Rechner nebeneinander stellen und vergleichen. Dabei wird die Vorgehensweise der Simulation gezeigt. Sehr ausführlich werden auch Annahmen, Vernachlässigungen und Erfahrungen bereitgestellt.

Zunächst wird das ANSYS-APDL-Programm verwendet. Am Ende des Beispiels wird für eine der Varianten das ANSYS/Workbench-Programm zur Lösung verwendet. Bei den Anwendungen des ANSYS-APDL-Programms wird eine Schreibweise verwendet, die am Anfang von Teil III und Teil IV näher erläutert ist. Bitte sehen Sie in diesen Erläuterungen nach, wenn unklar ist, wie die Darstellungen der Arbeitsschritte mit dem Programm zu verstehen sind.



Abb. 1 Lineal an der Tischkante




Abb. 2 Auslenkung des Lineals




Abb. 3 Schwingung des Lineals

1 Aufgabenstellung

Im Teil I des Buches waren die Definitionen dargestellt worden, die für die weiteren Abhandlungen über die Strukturdynamik wesentlich sind. Als Beispiel dabei war ein Lineal gezeigt worden, das an einer Tischkante festgehalten wird und als typischer Schwinger geschildert bezeichnet wurde.

Dieses schwingende Lineal soll nun mit dem Rechner simuliert werden. Das Lineal ist aus Kunststoff. Es ist etwas mehr als 30 cm lang, 4.3 cm breit und etwa 2 mm dick.

2 Idealisierung

2.1 Geometrie

Im Zuge der Idealisierung sind die Gegebenheiten der Aufgabenstellung in die Daten zu „übersetzen“, die zum Ausführen der Simulation benötigt werden. Bei dieser „Übersetzung“ muss man natürlich wissen, welche Daten benötigt werden, welche Rolle sie spielen und wie genau man diese Daten bestimmen und bereitstellen muss. Das entspricht der Frage:

„und was macht es aus, wenn ich mich hierbei doch verschätze?“ Also soll diese kritische Frage auch schon bei den Festlegungen und Annahmen über die geometrischen Abmessungen des Lineals gestellt werden.

Die Aufgabe der Berechnungen der Eigenschwingungen ist eine strukturmechanische Anwendung. Wir werden keine physikalischen Größen und Effekte betrachten, die mit thermischen, elektrischen oder sonstigen physikalischen Disziplinen zu tun haben. Bei dieser Entscheidung sind eigentlich keine nennenswerten Vernachlässigungen zu erkennen.

Das Lineal ist 3-dimensional räumlich ausgedehnt und kann daher als ein Volumenmodell behandelt werden. Es wäre zu überlegen, ob es wegen der dünnen und flachen Form auch wie ein Blech als ein Schalenmodell vereinfacht werden kann oder gar als schlankes längliches Bauteil als Balkenmodell abgebildet werden kann. Diese Vereinfachungen haben den Vorteil, dass der Rechner weniger zu rechnen hat. Sie haben den Nachteil, dass wir kaum Handhabungsaufwand sparen, wir einige Abstraktionen durchdenken müssen, die Ergebnisse nicht so einfach darzustellen und zu verstehen sind und einige Effekte vernachlässigt werden, die die Ergebnisse – die Eigenfrequenzen und Eigenformen – beeinflussen. Dies ergibt insgesamt: wir ziehen das Volumenmodell vor.

Das, was wir hier untersuchen wollen, ist eine Biegeschwingung des Lineals. Die Länge spielt eine mittel-wichtige Rolle, sie geht linear in die Steifigkeit ein. (Woher man das weiß? Das kann in Formelwerken und Literatur über Biegebalken mit Rechteck-Querschnitt nachvollzogen werden. Das wird hier nicht im Detail gemacht. Eine gute Möglichkeit zum Selbststudium!). Die Länge ist hier auch ganz gut auf einen Millimeter genau messbar. Sie ergibt sich zu $32\text{ cm} = 320\text{ mm}$. Die Restunsicherheit in Bezug auf die Frequenzen des Lineals schätzen wir auf wenige Prozent, damit geben wir uns zufrieden.

Auch die Breite geht linear in die Steifigkeit ein. Sie ist ein Bestandteil des Flächenträgheitsmomentes, das als maßgebende Größe für die Biegesteifigkeit zugrunde liegt. Für einen Rechteckquerschnitt kann man dessen Formel auch in Formelwerken und Literatur finden. Auch die Breite wird hier mit dem Lineal direkt gemessen, es hat ja eine Millimeter-Skala am Rand eingezeichnet, und es müssen nur zwei Striche an beiden Kanten des Lineals gezeichnet werden, und mit dem selben Lineal kann die Breite abgelesen werden. Die Breite kann damit auch auf einen Millimeter genau festgestellt werden mit $4.3\text{ cm} = 43\text{ mm}$ Abstand der seitlichen Kanten, für uns hier zufrieden stellend. Die Schrägen an den Kanten des Lineals berücksichtigen wir angenähert dadurch, dass wir von der Breite des Lineals einen kleinen Betrag abziehen, der eine mittlere Breite des Lineals repräsentiert. Damit wird die Breite zu $3.9\text{ cm} = 39\text{ mm}$ festgelegt. Die Restunsicherheit wird dadurch auch nur wenige Prozent betragen.

Die Dicke des Lineals ist ein besonderer Fall. Denn erstens ist sie ziemlich gering (etwa 2 mm) und damit schlecht messbar. Und zweitens geht sie in das Flächenträgheitsmoment des Rechteckquerschnitts in der 4. Potenz ein, also als Dicke hoch 4. Da hat eine kleine Messungenauigkeit schon eine große Auswirkung. Zum Beispiel: wenn die Dicke statt 2 mm nur 1.9 mm beträgt, macht dieser Unterschied von 5 % bei dem Trägheitsmoment schon mehr als 18 % aus, bei den Frequenzen allerdings wiederum nur knapp 10 %. Wie kann man die Dicke möglichst genau messen? Mancher hat einen Messschieber (früher eine Schieblehre genannt) zur Hand. Der




Abb. 4 Dickenmessung

Glückliche darf zum nächsten Absatz vorgehen. Ansonsten können wir ein Brett, ein anderes Lineal oder einen sonstigen länglichen Gegenstand mit einer geraden Kante nehmen und den Strahlensatz ausnutzen. In Abb. 4 ist dies skizziert: ein Brett mit 20 cm Länge, bei 3 cm ein Strich gemacht, schräg auf unser Lineal gehalten. Am Ende des Brettes messen wir 1.3 cm = 13 mm. Die Dicke des Lineals ist also $13 * 3 / 20 = 1.95$ mm. Das ist eine Genauigkeit, die hier ausreichen soll. Soweit die Anmerkungen zur Geometrie des Lineals.

2.2 Materialwerte

Nun zu den Materialwerten. Für die Steifigkeit des Lineals ist der Elastizitätsmodul ausschlaggebend. Wenn man eine Werkstoffbezeichnung hat oder im technischen Alltag das Material von einem Lieferanten eingekauft hat, kann man sich diesen Wert besorgen. Wir haben hier aber nur das Lineal vor uns, und der Aufwand zum Festlegen des Elastizitätsmoduls soll sich in Grenzen halten. Darum wird hier ein kleines Vor-Experiment durchgeführt und ein Abgleich mit einer FEM-Berechnung.

Legen Sie das Lineal an die Tischkante wie in Abb. 1 gezeigt. Wir haben hierbei die letzten 40 mm auf dem Tisch belassen, 280 mm ragen also über die Tischkante hinaus. Legen Sie jetzt auf das freie Ende ein Gewicht, zum Beispiel ein paar Münzen. Wir haben 2 1-Euro-Münzen genommen. Auf der Küchenwaage werden für 10 Münzen 75 g angezeigt (natürlich wird eine Gewichtskraft gemessen, mit der Erdbeschleunigung von 9.18 m/s^2 in eine Masse umgerechnet und als Masse angezeigt). Mit diesen 2 Münzen auf dem freien Ende messen wir eine Verbiegung nach unten von 10 mm.

Dieser Versuch wird nun mit dem ANSYS-APDL-Programm nachvollzogen, um den Elastizitätsmodul damit zu bestimmen. Hier wird nur kurz die Kommando-Folge aufgelistet. Sie wird nach dem Lesen von „FEM für Praktiker“, Band 1, nachvollziehbar sein. Ansonsten werden in den folgenden Teilen dieses Beispiels die für die Strukturtechnik wichtigen Eingaben ausführlicher beschrieben. Hier also nur so kurz wie möglich die Eingabe dieses statischen Modells:

/PREP7	
ET,1,45	! Elementtyp
BLOCK,0,0.32,0,0.039,0,0.00195	! die quaderförmige Geometrie
ESIZE,0.01	! Netzdichte
VMESH,ALL	! Vernetzen der Geometrie
MP,EX,1,1E9	! Elastizitätsmodul
MP,NUXY,1,0.3	! Querkontraktion
/SOLU	! Start des Lösungs-Abschnittes
NSEL,S,LOC,X,0,0.04	! im Bereich der Festhaltung
D,ALL,ALL	! alles festhalten
ALLS	! Selektieren aufheben
NSEL,S,LOC,X,0.3,0.32	! Bereich der Last am freien Ende
NSEL,R,LOC,Z,0.00195	! an der Oberseite
SF,ALL,PRES,0.153/(0.02*0.039)	! mit Flächenlast belasten
ALLS	! Selektieren aufheben
SOLVE	! Lösung durchführen
/POST1	! Start des Postprocessing
/VIEW,,-3,2,1	! Grafikeinstellung: Blickrichtung

**/VUP,,Z
PLDI,2**

**! Grafikeinstellung: z nach oben
! Darstellung der Verschiebungen**

Bei dieser Berechnung sind die oben diskutierten geometrischen Abmessungen des Lineals eingesetzt worden. Die Vernetzung ist so eingestellt worden, dass in Längsrichtung ein Element 20 mm lang ist. Dies erleichtert die Eingabe der Festhaltung: sie wird genau auf die Ecken der ersten beiden Elementreihen aufgebracht, also auf einen Abschnitt von 40 mm in Längsrichtung. Es wird der Elastizitätsmodul zunächst geschätzt auf einen Wert von 10^9 N/m². Die Verschiebungen werden zeigen, wie er zu korrigieren ist. Die Last, die die beiden 1-Euro-Münzen am Ende auf das Lineal ausüben, wird als flächig verteilte Last auf die letzte Elementreihe aufgebracht. Diese Flächenlast ergibt sich aus der Masse von 15 g = 0.015 kg, umgerechnet in eine Kraft von 0.153 N. Diese wird verteilt auf die Oberfläche der Elementreihe mit den Abmessungen 0.020 und 0.039 m.

Nach dieser Berechnung ergibt sich am Ende eine Verschiebung von 0.0433 m = 43.3 mm. Dies ist etwa um den Faktor 4 zu viel. Der reale Elastizitätsmodul muss also eher mit $1e9 * 43.3 / 10 = 4.33 * 10^9$ eingesetzt werden. Mit diesem Wert in die Berechnung eingesetzt ergibt sich tatsächlich eine Verschiebung des freien Endes des Lineals von 10 mm.

Über die Annahme, die diesem kleinen Test-Beispiel zugrunde liegt, dass nämlich der Elastizitätsmodul und die Verschiebung am Ende linear miteinander in Beziehung stehen, wird im Folgenden noch beraten.

Die Querkontraktion ist bei den meisten technischen Werkstoffen etwa 0.3, also in dem Wertebereich von 0.28 bis 0.33. Dieser Wert ist für die hier zu untersuchende Biegeschwingung von geringem Einfluss, so dass ohne weitere Detailuntersuchung ein Wert von 0.3 eingesetzt wird.

Wir müssen für die Berechnung der Strukturmechanik auch die Dichte des Materials kennen. Wir haben dafür das Lineal auf die Waage gelegt und 29 g als Anzeige abgelesen. Bei dem Volumen des Lineals ergibt sich daraus eine Dichte von 1192 kg/m³.

2.3 Sonstige Annahmen

Die bisherigen Schritte der Idealisierung enthielten bereits einige Restunsicherheiten, die wir bisher jeweils im Bereich weniger Prozente abgeschätzt haben. Auch die folgenden Annahmen entscheiden über Einflüsse in dieser Größenordnung.

Dazu zählt die Annahme des linearen Materialverhaltens. Die Dichte kann guten Gewissens als linear angesehen werden. Linear bei dieser Anwendung der Strukturmechanik heißt „nicht von den Verschiebungen (und damit auch den Geschwindigkeiten und Beschleunigungen) abhängig“. Und Geschwindigkeiten, bei denen nach Einstein diese Annahme in Frage gestellt werden könnte, erreichen wir nun wirklich nicht annähernd. Die Annahme eines linearen Verhaltens des Elastizitätsmoduls ist schon etwas mutiger. Diese Linearität ist zum Beispiel naheliegend, wenn man feststellt, dass das Lineal so verbogen werden kann, wie wir es hier tun, und es danach wieder die Ausgangsform einnimmt, also sich nicht bleibend verformt.

Eine weitere Annahme ist die einer linearen Schwingung. Das bedeutet, dass eine geringe Auslenkung und eine größere Auslenkung des schwingenden Lineals zueinander

proportional sind und beide Zustände direkt linear in Relation stehen. Dies trifft zum Beispiel dadurch nicht genau zu, dass der Weg des freien Endes des Lineals einen Kreisbogen beschreibt, während eine lineare Schwingung nur die vertikalen Bewegungen beobachtet und berücksichtigt. Die Annahme der linearen Schwingung wird um so mehr verletzt, je größer die Ausschläge des Lineals beim Schwingen sind.

Für die feste Einspannung werden alle Freiheitsgrade (DOF) der Knoten festgehalten. Festhalten im numerischen Sinne bedeutet, dass perfekt fixiert wird und absolut keine Verschiebung stattfindet. Das trifft auf die Finger, die das Ende des Lineals auf den Tisch pressen, sicherlich nicht zu.

Beim Schwingen tritt bei unserem FEM-Modell kein Luftwiderstand auf. In Realität ergibt sich dadurch ein Dämpfungseffekt und auch eine kleine Beeinflussung der Eigenfrequenzen.

Wir vernachlässigen das Eigengewicht des Lineals. Dabei können wir ein gutes Gewissen haben, denn in Teil I des Buches wurde gezeigt, dass das Eigengewicht „nur“ zu einer Verschiebung der Ruhelage des Schwingers führt und ansonsten keine Beeinflussung der Eigenfrequenzen zu erwarten ist.

Alle diese Annahmen, die hier bei der Idealisierung getroffen wurden, ließen Restunsicherheiten bestehen. Im technischen Alltag heben sich diese Restunsicherheiten oftmals auf, so dass die Simulation eine gute Treffsicherheit bekommt. Auch hier bei diesem Beispiel dürfte die erste Eigenfrequenz nicht weiter als 5 %, vielleicht sogar 2 % von der Realität entfernt sein. Natürlich soll auch erst einmal ein Messwert der Frequenz so genau sein, dass diese Schätzung angezweifelt werden kann!

Im technischen Alltag sind allerdings auch oft die Bedingungen gegeben, dass die Simulation ein Bauteil betrifft, das später dann millionenfach hergestellt wird. Bei solchen Stückzahlen ist es natürlich durchaus möglich, dass sich die zahlreichen Restunsicherheiten und Annahmen unglücklich addieren und ein Fall auftritt, bei dem das Ergebnis um 10 % oder 20 % von der Realität abweicht. Das sind dann Fälle, bei denen ein besonders dünn ausgefallenes Lineal und ein besonders weicher Werkstoff und andere extreme Bedingungen zufällig zusammen treffen. Das ist dann eine Simulationsaufgabe, die Wahrscheinlichkeiten und Stochastik mit in die Überlegungen einzubeziehen hat. Diese Vorgehensweisen bei der Simulation sind durchaus üblich. Sie werden aber nicht hier weiter vertieft.

3 Modellerstellung (preprocessing)

Für die Modellerstellung starten wir den Preprocessor mit **/PREP7**

Als erster Arbeitsschritt der Modellerstellung wird ein Elementtyp ausgewählt. Das wird ganz am Anfang der Anwendung gemacht, weil dem Programm dadurch einige wesentliche Entscheidungen aus der Idealisierung mitgeteilt werden wie:

- Ist es eine strukturmechanische oder eine thermische oder eine elektrische Simulation oder liegt eine sonstige physikalische Disziplin zugrunde?
- Ist das Modell 3-dimensional räumlich ausgedehnt wie das Lineal hier (Volumenmodell) oder ist es ganz dünn und flach wie ein Blech (Schalenmodell) oder gar aus

dünnen schlanken Bauteilen aufgebaut wie der Eiffelturm (Balkenmodell)?

Für unsere strukturmechanische Anwendung mit einem Volumenmodell wird als Elementtyp der Typ SOLID45 ausgewählt. Die Auswahl erfolgt mit

ET,1,SOLID45

Die Geometrie des Lineals ist quaderförmig. Als x-Richtung wird die Lineal-Längsachse, als y-Richtung die Querrichtung und als z-Richtung die Dickenrichtung verwendet. Der Ursprung der Koordinaten wird an eine Ecke des Lineals gelegt. Damit erstreckt sich das Volumen in x-Richtung von 0 bis 0.32 m, in y-Richtung von 0 bis 0.039 m und in z-Richtung von 0 bis 0.00195 m. Ein Quader mit diesen Abmessungen wird eingegeben mit

BLOCK,0,0.32,0,0.039,0,0.00195

Dieser Quader wird jetzt mit einer Vernetzung versehen. In diesem Beispiel soll das Netz halbwegs fein in Längsrichtung sein, hier ist eine Elementgröße von 10 mm eingesetzt. Die Berechnung läuft so schnell auf dem Rechner ab, da lohnt es sich nicht, zu geizen. Diese Vorgabe der Teilung wird eingegeben mit

ESIZE,0.01

Wenn diese Vorgabe nicht erfolgen würde, würden nur etwa 15 Elemente in Längsrichtung angeordnet werden. Mit dieser Vorgabe werden es aber über 60. Mit einem solchen Modell können wir gleich allen Diskussionen um die Elementteilung aus dem Weg gehen.

Mit dieser Einstellung wird die Vernetzung ausgeführt durch

VMESH,ALL

Jetzt werden die Materialwerte dem Programm mitgeteilt. Dies ist zunächst der Elastizitätsmodul

MP,EX,1,4.33e9

dann die Querkontraktion

MP,NUXY,1,0.3

und noch die Dichte

MP,DENS,1,1192

Dies sind die Eingaben, die zur Festlegung des Modells erforderlich sind.

4 Aufbringen der Lasten und Starten der Lösung (solution)

Für die Vorbereitung der Lösung werden nun einige Steuergrößen eingegeben. Außerdem werden die von außen auf das Modell einwirkenden Effekte (Randbedingungen) eingegeben. Dann wird auf den Lösungs-„knopf“ gedrückt, so dass der Rechner die eigentliche Zahlenverarbeitung starten kann.

Der Beginn des Lösungs-Abschnittes des Programms wird eingeleitet durch

/SOLU

Gleich am Anfang dieses Abschnittes muss dem Programm unsere Aufgabe als Modalanalyse deklariert werden. Davon sind viele Schalterstellungen der anschließenden Eingaben abhängig. Darum ist diese Eingabe im Menü des Lösungsabschnittes ganz oben angeboten, die Auswahl muss auch gleich am Anfang erfolgen. Hier erfolgt diese Auswahl durch

ANTYPE,MODAL

Innerhalb der Modalanalyse sind weitere Detailangaben zu treffen. Dazu zählt die Auswahl der Berechnungsmethode, hier empfiehlt die Beschreibung des Programm die Block-Lanczos-Methode, auszuwählen durch die Kennung LANB. Dieser Empfehlung wird hier entsprochen. Außerdem müssen wir die Anzahl der Eigenwerte angeben, die wir berechnet haben möchten. Bei dem Lineal interessiert uns im wesentlichen die 1. Eigenfrequenz und Eigenform, aber der Rechner hat keine Probleme, auch ein paar weitere Ergebnisse zu bestimmen. Darum wird hier 4 eingegeben, damit später anhand der Ergebnisse noch weitere dynamische Erfahrungen diskutiert werden können. Die Eingabe hier erfolgt also mit **MODOPT, LANB, 4**

Jetzt kümmern wir uns um die Belastungen des Modells. Der Begriff lässt zunächst nur an Kräfte und Momente denken, aber hiermit sind alle von außen einwirkenden Einflüsse gemeint, also auch die Einflüsse von Lagerungen und Festhaltungen. Hier geben wir die Daten ein, die die Festhaltung des Lineals auf dem Tisch darstellen. Diese Festhaltung wird dadurch erreicht, dass wir die ersten 40 mm der Länge des Lineals auf dem Tisch liegen lassen und so fest mit der Hand darauf drücken, dass keine Verschiebung auftreten kann. Es war schon vorher darauf hingewiesen worden, dass diese Beschreibung zu mutig ist, alle Verschiebungen wird unsere Hand nicht vollständig unterdrücken können. Aber im Sinne unserer Berechnungsaufgabe ist die „vollständige“ Festhaltung genau genug zutreffend. Mit der Dateneingabe werden zunächst die betreffenden Bereiche des Lineals ausgewählt (selektiert) über ihre Eigenschaft, in x-Richtung im Bereich von 0 bis 0.04 m zu liegen, so dass die Eingabe erfolgt mit

NSEL,S,LOC,X,0,0.04

Die Knoten in diesem Bereich werden komplett festgehalten und fixiert mit

D,ALL,ALL

und die Auswahl wieder aufgehoben mit

ALLS

Mit diesen Dateneingaben sind alle notwendigen Angaben dem Programm übermittelt, so dass die Lösung angefordert werden kann mit

SOLVE

Die Ausgabe während der Lösung, die in dem „Output“-Fenster erfolgt, beinhaltet eine Anzahl von Meldungen über den Berechnungsfortschritt und am Ende bereits die Eigenfrequenzen. Ungeduldige Anwender können also dieses Fenster bereits ansehen und die Zahlenwerte der Ergebnisse ablesen. Ansonsten ist aber vorgesehen, dass man die berechneten Ergebnisse ausführlich und im Detail ansehen möchte. Dies erfolgt in dem nächsten Arbeitsabschnitt, dem Postprocessing.

5 Auswertung der Ergebnisse (postprocessing)

Die Arbeit mit dem allgemeinen Postprocessor wird begonnen mit der Eingabe

/POST1

Die berechneten Eigenfrequenzen werden aufgelistet mit

SET,LIST

Die Liste zeigt die Einträge

SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE
1	7.7519	1	1	1
2	48.668	1	2	2
3	110.02	1	3	3
4	137.27	1	4	4

In jeder Zeile der Liste steht als „SET“ eine Nummer der Frequenz, dann der Frequenzwert und danach Angaben darüber, als welcher „LOAD STEP“ und „SUBSTEP“ die zugehörigen Ergebnisse aus den gespeicherten Daten abgerufen werden kann. Denn für jede Eigenfrequenz ist die zugehörige Eigenschwingungsform auf der Ergebnisdatei als separater Zwischenschritt abgespeichert. Um die Daten des 1. Modes zu lesen, verwenden wir den *SBSTEP*-Parameter von

SET,,1

Die grafische Darstellung der Eigenschwingungsform kann jetzt mit

PLDISP,2

erfolgen. **PLDISP** stellt die aktuelle, in der Datenbasis vorliegende Eigenform dar. Der *KUND*-Parameter von **PLDISP** ergibt eine Darstellung der Verformung zusammen mit einer überlagerten Darstellung der Umrisse der unverformten Struktur. Die Zahlenwerte der Ergebnisse, die zu den Eigenformen gehören, sind normiert und haben nicht direkt etwas mit einer reale zu erwartenden Verformung zu tun, sie repräsentieren nur die „typische“ Form dieser Eigenschwingung. Drehen Sie das Bild auf dem Bildschirm so, wie Sie es am besten mit dem Foto des Lineals vergleichen können.

Mit einer Animation kann direkt als Bildfolge das in der 1. Eigenform schwingende Lineal dargestellt werden

ANMODE,10,.1

Die weiteren Eigenformen können entsprechend dargestellt werden. Für die 2. Eigenfrequenz und die zugehörige Eigenform sind es die Eingaben

SET,,2

PLDISP,2

ANMODE,10,.1

Für die 3. Eigenfrequenz und ihre Eigenform verwenden Sie

SET,,3

PLDISP,2

ANMODE,10,.1

Und für die 4. Eigenfrequenz und die zugehörige Eigenform sind es die Eingaben

SET,,4

PLDISP,2

ANMODE,10,.1




Abb. 5 1. Eigenform

Jetzt bleiben einige Punkte über die Eigenfrequenzen zu diskutieren. Der Vergleich zwischen Experiment und Berechnung wurden hier für die 1. Eigenfrequenz gemacht. Warum nicht auch für die anderen Eigenfrequenzen? Was hat es mit diesen Eigenfrequenzen auf sich?

Die Eigenfrequenzen, die die Modalanalyse liefert, sind mögliche Schwingungen dieses Lineals, aber keine real auftretenden Schwingungen, jedenfalls nicht real auftretend, wenn die Bedingungen vorliegen, die hier für die Simulation angegeben worden sind. Und dabei ist das Wichtigste, dass hier für die Simulation nur das Bauteil und seine Einspannbedingungen angegeben wurden, aber überhaupt keine Belastungen, die das Lineal aus der Ruhelage gebracht hätten. Die Simulation hat also nur Schwingungen berechnet, die das Lineal ausführen würde, wenn.... Und diese „wenn“ lauten zum Beispiel:

- wenn eine entsprechende Last vorhanden ist und
- diese Last diese Eigenfrequenz und Eigenform auch anregen kann.

In dieser Hinsicht ist der Vergleich der 1. Eigenform und des Experimentes mit dem schwingenden Lineal in Abb. 1 tückisch gewesen, es war sozusagen ein Vergleich zwischen Äpfeln und Birnen, denn:

- die Darstellung der 1. Eigenform aus der Simulation ist eine Schwingung, zu der das Lineal in der Lage ist (aber nur, wenn bestimmte Lasten vorhanden sind oder waren) und
- die Darstellung des Experimentes mit dem schwingenden Lineal ist ein Ausschnitt aus einem zeitlichen Geschehen, bei dem wirklich solche Lasten wirken oder gewirkt haben (und bei der Darstellung kann auch diskutiert werden, ob wir wirklich die 1. Eigenform sehen oder noch was anderes dabei).




Abb. 6 Eigenfrequenzen, Eigenformen

Die Darstellungen der 1. bis 4. Eigenform in Abb. 6 erwähnen die 1. und die weiteren Eigenformen als

- 1. Biegung,
- 2. Biegung,
- Verdrillung und
- 3. Biegung.

Wie die 1. Eigenform sind die weiteren Eigenformen die Verformungen, die auftreten können, wenn entsprechende Lasten vorhanden sind oder waren. Die Simulation behauptet also, dass das Lineal durchaus so schwingen kann wie als 2. Biegung oder Verdrillung oder 3. Biegung bezeichnet. Wie muss ein Experiment aussehen, dass eine solche Schwingung liefert?

Jetzt wird es schwierig, denn solche Experimente sind mit dem Lineal nicht mehr sinnvoll am Küchentisch mit Hausmitteln auszuführen. Es sind andere Bauteile als Lineale denkbar (zum Beispiel ein langes Brett, das sehr viel langsamer und träger schwingt als das kleine Lineal), es sind Video-Kameras denkbar, um genauer hinsehen zu können (hat nicht jemand eine Digital-Kamera, deren Bilder dann auf dem Rechner Schritt für Schritt untersucht werden können? Natürlich ist die Aufnahme von schnellen Bewegungen oft verwischt...), und andere hochtechnische Apparate, aber diese Ausrüstung ist jenseits dessen, was wir hier in unserem Physik-Lab für möglich halten.

Was man aber machen könnte, ist eine Simulation von dem Geschehen. Der Rechner und das ANSYS-Programm sind genau richtig, um auf dem Bildschirm zu simulieren, was wäre, wenn wir die Ausrüstung hätten.

6 Zeitverlaufs-Berechnungen

Unsere Behauptung ist, dass wir bei dem Experiment bisher nichts außer der 1. Eigenfrequenz beobachten konnten, weil sie von den Amplituden her so dominant ist und weil die bisher gewählte Belastung auch kaum die anderen Eigenfrequenzen anregen konnte. Das Gegenteil im Experiment zeigen ist wirklich schwierig. Hier wird jetzt nur der einfache Weg gewählt, nämlich die Simulation durchzuführen und anhand der Ergebnisse diese Behauptung zu untermauern. Anschließend kann der Leser selber überlegen, ob vielleicht eine Erweiterung des Experimentes oder ein anderes Experiment sinnvoll ist.

Die hier folgenden Simulationen haben einen doppelten Sinn. Sie werden

1. dazu dienen, das Experiment zu ersetzen und zu simulieren und
2. eine andere Berechnungsmethode als bisher verwenden.

Die bisherige Berechnungsmethode war die Modalanalyse. Sie hat sich auf die Frequenzen konzentriert. Jetzt wird die direkte Zeitintegration verwendet, die sich mit dem zeitlichen Verlauf der strukturmechanischen Größen beschäftigt.

6.1 Variante 1: Last am Ende

Und in der Simulation:

Modellerstellung wie oben

```
/PREP7
ET,1,SOLID45
BLOCK,0,0.32,0,0.039,0,0.00195
ESIZE,0.01
VMESH,ALL
MP,EX,1,4.33E9
MP,NUXY,1,0.3
MP,DENS,1,1192
```




Abb. 7 Variante 1:
Lastangriffspunkte

Damit die Weiterarbeit erleichtert wird, speichern Sie diese Variante mit

```
SAVE,LINEAL1,DB
```

Im Lösungsabschnitt wird hier der Analysetyp „transient“ für die Zeitintegration eingesetzt

```
/SOLU
```

```
ANTYPE,TRANS
TRNOPT,FULL
```

! Festhaltung

```
NSEL,S,LOC,X,0,0.04
D,ALL,ALL
ALLS
```

Zunächst wird die statische Auslenkung berechnet. Dabei wird auf die Nachrechnung des zeitlichen Verlaufes für diese Auslenkung verzichtet. Die dynamischen Anteile der Masse (und Dämpfung, ist hier aber nicht im Modell enthalten) werden inaktiviert. Dieser Lastfall entspricht damit einem statischen Lastfall.

! Statische Auslenkung

```
TIME,1
DELTIM,1
TIMINT,OFF
```

! Last am freien Ende

```
NSEL,S,LOC,X,0.3,0.32
NSEL,R,LOC,Z,0.00195
```

! auf $0.02 \cdot 0.039$ sollen

! etwa $0.015 \cdot 9.81$ N wirken

```
SF,ALL,PRES,0.153/(0.02*0.039)
ALLS
SOLVE
```

Von diesem Zustand ausgehend wird nun die Last entfernt. Das entspricht dem Loslassen des Lineals.

! Loslassen

```

TIMINT,ON
TIME,1.5
AUTOTS,ON
DELTIM,0.005,0.001,0.01
OUTRES,ALL,ALL
SFDEL,ALL,PRES
SOLVE

```

Das Ergebnis ist der zeitliche Verlauf der Verformungen und anderer mechanischer Größen. Diese Ergebnisgrößen sind bei der Lösung berechnet worden, sie stehen jetzt zur Auswertung zur Verfügung.

Hier in diesem Beispiel werden die Verschiebungen am Ende des Lineals untersucht und als Kurvenverlauf über der Zeit dargestellt. Die beiden Ecken des Lineals, deren Verformungen dargestellt werden, sind in Abb. 7 gekennzeichnet. Der Zeitverlauf wird dargestellt mit

```

/POST26
NSOL,2,NODE(0.32,0,0.00195),U,Z
NSOL,3,NODE(0.32,0.039,0.00195),U,Z
PLVAR,2,3

```

Das Ergebnis ist als Kurvenverlauf in Abb. 8 zu sehen. Beide Ecken bewegen sich so genau gleich, dass sie auf der grafischen Darstellung nicht zu unterscheiden sind. Trotzdem wurden hier beide Ecken ausgewertet, und zwar einmal, um zu beweisen, dass sie in dieser Variante gleiche Verschiebungen ausführen, und außerdem, weil in den Varianten, die jetzt noch folgen, nicht immer gleiche Verschiebungen ausführen. Hier ist die Bewegung des Lineal-Endes eine schöne saubere Schwingung. Sie beginnt bei $t = 1$ s und hat beim Zeitpunkt $t = 1.39$ s – also nach 0.39 s - 3 Schwingungen ausgeführt. Das entspricht genau der 1. Eigenfrequenz mit 7.7 Hz bzw. einer Periodendauer von 0.13 s.




Abb. 8 Zeitverlauf 1. Biegung

Können wir also feststellen, dass die gewählte Anfangs-Belastung nur die 1. Eigenschwingung anregt? Ingenieurmäßig kann dies mit "ja" beantwortet werden, und für eine technische Aufgabenstellung wäre dieser Zeitverlauf sicherlich ausreichend. Ganz korrekt und genau können wir das aber nicht so ohne weiteres sagen, denn andere Eigenschwingungen können enthalten sein, nur vielleicht in so geringem Maße, dass sie vernachlässigt werden können. Das werden wir bei den nächsten Varianten noch sehen.

6.2 Variante 2: Last bei 1/3 der Länge

In Variante 2 wird diese Anwendung wiederholt, aber die Lastaufbringung etwas abgeändert. Ein Experiment haben wir dazu nicht vorbereitet, darum wird hier nur die Simulation und ihre Ergebnisse gezeigt.

Das Lineal wird weiterhin so an der Tischkante festgehalten wie bisher. Am Anfang wird aber nicht das freie Ende des Lineals ausgelenkt und nach unten gedrückt, sondern eine Stelle, die etwa bei 1/3 der Länge des Lineals liegt. Diese Stelle ist in Abb. 9 eingezeichnet.




Abb. 9 Variante 2:
Lastangriffspunkte

Das so nach unten gebogene Lineal wird dann plötzlich losgelassen, so dass es frei ausschlagen kann. Diese Auslenkung und die danach stattfindende Bewegung wird jetzt simuliert.

Im ANSYS-APDL-Programm wird dazu die letzte Variante wieder geladen mit **RESUME,LINEAL1,DB**

und die Änderung vorgenommen. Die Änderung betrifft die Last, die am Anfang das Lineal verformt. Sie ist jetzt bei 1/3 der freien Länge des Lineals angeordnet. Das wird dem Programm eingegeben mit den Zeilen

```

FINI
/SOLU
ANTYPE,TRANS
TRNOPT,FULL
TIME,1
DELTIM,1
TIMINT,OFF
! LAST BEI 1/3
NSEL,S,LOC,X,0.1,0.12
NSEL,R,LOC,Z,0.00195
SF,ALL,PRES,0.153/(0.02*0.039)
ALLS
SOLVE
! Loslassen
TIMINT,ON
TIME,1.5
AUTOTS,ON
DELTIM,0.005,0.001,0.01
OUTRES,ALL,ALL
SFDEL,ALL,PRES
SOLVE
.....
    
```




Abb. 10 Zeitverlauf 1. und 2. Biegung

Ansonsten bleibt die Vorgehensweise gleich. Nach der Berechnung werden die berechneten Ergebniswerte im Postprocessing ausgewertet.

Die Kommandoingabe ist im wesentlichen identisch zu derjenigen der Variante 1. In diesem Fall lohnt es sich, die Zeilen in einer Textdatei im Editor notiert zu haben und sie von dort in das Kommandofenster eingefügt zu haben. Dann kann eine solche Variante schnell durchgeführt werden.



Es kann auch das vergrößerte Eingabefenster verwendet werden, um die vorher bereits einmal eingegebenen Kommandos wieder anzuklicken und auszuführen. Bei dieser Vorgehensweise beschränkt sich der Aufwand im wesentlichen nur noch auf das Ausführen von Änderungen gegenüber der vorigen Variante.

Der Zeitverlauf der Verschiebungen am freien Ende des Lineals sind in Abb. 10 dargestellt. Es ergibt sich ein zeitlicher Verlauf, der eine Mischung aus 1. und 2. Biegung darstellt. Die Auslenkung am Anfang war so gewählt worden, dass beide Eigenfrequenzen angeregt wurden. Vorher, bei Variante 1, war die Biegeform des

Lineals sehr ähnlich zu derjenigen der 1. Eigenform. Dies ist in Abb. 11 oben skizziert. Hier, in Variante 2, ist die Biegeform durch die Last bei 1/3 der Länge gekennzeichnet, bis hier findet die Biegung statt, daran anschließend wird die Schrägstellung des Lineals durch die restliche Länge nur noch als eine Gerade fortgesetzt. Wenn das so verbogene Lineal losgelassen wird, schwingt es sowohl in der 1. Eigenfrequenz als auch in höheren Eigenfrequenzen. Der Zeitverlauf zeigt, dass die Anteile der 1. Biegung




Abb. 11 Vergleich der Lastangriffspunkte

(1. Eigenform) vorherrschend sind, sie sind vergleichbar zu dem Zeitverlauf der vorigen Variante 1. Die Anteile der 2. Biegung (2. Eigenform) sind nur als leichte Störung der harmonischen (Sinus-) Kurve des Anteils der 1. Frequenz erkennbar. Diese 2. Biegung ist höherfrequent, dadurch ergeben sich - selbst wenn beide Eigenformen vergleichbare Energieinhalte hätten - für die höherfrequente Schwingung geringere Amplituden.

Eine interessante Untersuchung kann der Leser bei dieser Variante selber durchführen: verschieben Sie den Lastangriffspunkt nicht auf 1/3 der Länge, sondern auf 2/3 der Länge. Was für einen Zeitverlauf erwarten Sie? Es ergibt sich nur eine deutliche Biegung in der 1. Eigenfrequenz und ihrer Eigenform, aber keinen erkennbaren Anteil der 2. Eigenform. Warum? Weil am Lastangriffspunkt ein Schwingungsknoten der Eigenform der 2. Biegung ist. Daher wird diese Eigenfrequenz nicht angeregt. Dies ist ein typisches Beispiel für die schwierige Aufgabenstellung, wenn Versuche zur Eigenfrequenzbestimmung durchgeführt werden sollen. Häufig nimmt man dafür im technischen Alltag das Bauteil, befestigt eine Anzahl Beschleunigungsaufnehmer (Sensoren) daran und schlägt leicht mit einem Hammer an die geeigneten Stellen. Damit werden die Schwingungen angeregt. Das Signal der Sensoren wird ausgewertet und daraus mit einer Fourier-Analyse auf die Eigenfrequenzen geschlossen. Außerdem kann man aus einem Vergleich der Signale der einzelnen Sensoren

auch Einiges über die Eigenformen herausbekommen. Bei der Planung dieses Versuches muss man die Schwingungsbüchse der Eigenformen, die man messen will, aber schon vorher ahnen, denn dort muss man anregen und mit dem Hammer draufschlagen. Wenn man einen Ort wählt, wo eine gesuchte Schwingung einen Schwingungsknoten hat, kann man die entsprechende Eigenform nie anregen und entsprechend auch nicht experimentell auffinden. Bei solchen Aufgabenstellungen ist es sinnvoll, die Simulation und den Versuch zu kombinieren, um mit der Simulation eine Abschätzung der Eigenform zu erhalten und mit dem Versuch die Annahmen der Simulation abzusichern.

6.3 Variante 3: Last an der Kante

Bei unserem Lineal ist als 3. Eigenfrequenz eine Verdrillung berechnet worden. Kann sie auch beobachtet und angeregt werden?

Ein Experiment hierzu ist vorstellbar. Wir haben es zwar durchgeführt, aber nicht ausgewertet, weil die dazu erforderlichen Apparate zum Feststellen der hochfrequenten Anteile nicht erfügar waren. Darum muss auch diese Variante sich auf die Simulation beschränken.

Die Simulation verwendet wie bei den vorigen Varianten nur eine geänderte Lasteinleitung für die Anfangsauslenkung. Hier wird die Last als Druck entlang der Außenkante einer Seite am freien Ende aufgebracht. Dazu wird in der Simulation ein Streifen der Modell-Oberseite belastet, der in Längsrichtung von 0.28 bis 0.32 m reicht (40 mm Länge) und in Querrichtung nur 0.01 m (10 mm) breit ist. Dieser Streifen liegt an der Kante des Lineal-Modells. Die entsprechenden Dateneingaben hierfür lauten

```
.....
TIMINT,OFF
! Last aussen
NSEL,S,LOC,X,0.28,0.32
NSEL,R,LOC,Y,0,0.01
NSEL,R,LOC,Z,0,0.00195
SF,ALL,PRES,0.153/(0.04*0.01)
ALLS
.....
```

Der Zeitverlauf, der hiermit berechnet wird, ist in Abb. 13 gezeigt. Es ist auf den ersten Blick eine reine Biegung mit der 1. Eigenfrequenz. Erst auf den zweiten Blick ist zu erkennen, dass es eine Kombination von Biegung und Verdrillung ist. Aber die Amplituden der Verdrillung sind so gering, dass sie nur als ganz leichte Abweichung der beiden Verschiebungen der äußeren Ecken feststellbar ist. Im




Abb. 12 Variante 3:
Lastangriffspunkte




Abb. 13 Zeitverlauf 1. Biegung und Verdrillung

Vergleich der Liste der Zahlenwerte, die in Abb. 13 dargestellt sind, ist diese Verdrillung erkennbar. Die Liste enthält im Bereich des ersten Maximums bei etwa $t = 1.07$ s die Zahlenwerte

	UZ	UZ
1.0000	-0.922640E-02	-0.909415E-02
1.0050	-0.901192E-02	-0.898271E-02
1.0067	-0.882015E-02	-0.885124E-02
1.0083	-0.858276E-02	-0.863790E-02
1.0120	-0.796908E-02	-0.795745E-02
1.0157	-0.718534E-02	-0.713450E-02
1.0252	-0.430677E-02	-0.436191E-02
1.0348	-0.391841E-03	-0.350905E-03
1.0443	0.377615E-02	0.376298E-02
1.0539	0.680105E-02	0.678230E-02
1.0634	0.855433E-02	0.859856E-02
1.0730	0.883214E-02	0.877718E-02
1.0762	0.837882E-02	0.838984E-02
1.0793	0.767231E-02	0.772195E-02
1.0862	0.569199E-02	0.564783E-02
1.0931	0.326879E-02	0.328068E-02
1.0999	0.528943E-03	0.556114E-03

Hier sind in der linken Spalte die Zeitwerte, in der mittleren Spalte die Verschiebungen der einen Ecke des Lineal-Endes und in der rechten Spalte diejenigen der anderen Ecke des Lineal-Endes aufgelistet. Vergleichen Sie die mittlere und die rechte Spalte. Es ist immer mal der linke Wert größer, dann wieder der rechte Wert. Dieses Verhalten zeigt, dass die beiden Ecken gegeneinander schwingen. Diese Verdrillschwingung ist aber mit nur ganz geringen Amplituden beteiligt. Die Amplituden sind so gering, dass die Zeitschritt-Automatik des Programms, die hier verwendet wird, darauf keine Rücksicht nimmt. Im technischen Alltag wäre also dieser Anteil der Verdrillung als unbedeutend eingestuft und vernachlässigt worden.

6.4 Variante 4: Last an beiden Kanten

Kann man vielleicht diese Verdrillung gar nicht simulieren? Doch, das kann man. Und zwar wird dies in dieser Variante 4 dadurch erreicht, dass wir die Lasteinleitung so gestalten, dass die Verdrillung deutlich und vorherrschend angeregt wird. Dazu wird eine Lasteinleitung gewählt, bei der ein Druck an der einen Kante wie in der letzten Variante 3 aufgebracht wird, aber zusätzlich ein Gegendruck auf der anderen Seite. Dadurch wird ein verdrillendes Moment am Ende des Lineals eingeleitet. Auch dies ist experimentell vorstellbar, wir haben es aber nur als Simulation vorbereitet. Die entsprechenden Eingaben lauten dann




Abb. 14 Variante 4:
Lastangriffspunkte

```
.....
TIMINT,OFF
! LAST AUSSEN
NSEL,S,LOC,X,0.28,0.32
NSEL,R,LOC,Y,0,0.01
NSEL,R,LOC,Z,0.00195
```

```
SF,ALL,PRES,0.153/(0.04*0.01)
NSEL,S,LOC,X,0.28,0.32
NSEL,R,LOC,Y,0.029,0.039
NSEL,R,LOC,Z,0
SF,ALL,PRES,0.153/(0.04*0.01)
ALLS
SOLVE
```

```
TIMINT,ON
TIME,1.05
AUTOTS,ON
DELTIM,0.002,0.0002,0.01
OUTRES,ALL,ALL
SFDEL,ALL,PRES
SOLVE.....
```




Abb. 15 Zeitverlauf Verdrillung

Dies ergibt wie vorher eine Last auf einen Streifen der Modell-Oberseite (40 mm Länge, 10 mm breit), der an der Kante des Lineal-Modells liegt. Zusätzlich wird eine gleiche Last auf einen Streifen mit den gleichen Abmessungen an der Unterseite an der anderen Kante des Modells aufgebracht.

Außerdem wurde der zu berechnende Zeitbereich und die Zeitschrittweite geändert, damit die kleineren Perioden der 3. Eigenfrequenz aufgelöst und erkennbar werden.

Das Ergebnis ist als Kurvenverlauf in Abb. 15 zu sehen. Die beiden Eckpunkte des freien Endes des Lineals schwingen gegeneinander. Die Schwingung beginnt bei $t = 1$ s und hat beim Zeitpunkt $t = 1.047$ s – also nach 0.047 s - 5 Schwingungen ausgeführt. Das entspricht recht genau der 3. Eigenfrequenz mit 108.7 Hz bzw. einer Periodendauer von 0.0092 s.

7 Varianten zum Selbststudium

7.1 Masseverteilung

Bei diesem Beispiel liegt mit dem Lineal ein Schwinger mit verteilter Masse vor. Das ist eigentlich ein schwieriges Bauteil, das in die Gruppe der n-Massenschwinger einzustufen ist. Es müsste sogar der Grenzfall "n gegen unendlich" betrachtet werden.

Wir konnten trotzdem ganz gut Experimente machen und Simulationen durchführen, weil hier die ersten Eigenfrequenzen und Eigenformen deutlich auftraten und leicht zu erklären waren. Aber bereits eine Vergleichsrechnung per Hand ist nicht mehr einfach, weil hierfür diese verteilte Masse zu idealisieren ist.




Abb. 16 Varianten der Masseverteilung

Im technischen Alltag ist durchaus für eine erste grobe Abschätzung der 1. Eigenfrequenz eine Untersuchung mit einer Aufteilung der Masse üblich. Dabei wird die Gesamtmasse (in

Abb. 16 oben als breiter Strich skizziert) auf die beiden Enden des Schwingers aufgeteilt und dort anteilig angeordnet. Damit ergibt sich ein Hantel-artiges Gebilde wie in Abb. 16 in der Mitte angedeutet. Ein solches Bauteil liegt ja auch schon etwa vor, wenn das Lineal am Ende mit schweren Gegenständen wie Münzen oder Schrauben verbunden wird, so dass diese Zusatzmassen eine größere Rolle spielen und die verteilte Masse des Kunststoff-Lineals eine geringere Rolle spielt. Den idealen Endzustand, der in Abb. 16 unten dargestellt ist, bei dem die Gesamtmasse am Ende des Schwingers angeordnet ist und in der Mitte die Steifigkeit durch ein masseloses Lineal geliefert wird, wird es ja nie geben können: denn welches Bauteil ist schon wirklich masselos? Dieses ideale Gebilde ergibt aber trotzdem eine erste grobe Näherung für unser Experiment und die Simulation mit dem Lineal.

Eine solche Handrechnung kann also die halbe Masse am Ende konzentriert annehmen. Die Masse ergibt sich aus dem Volumen und der Dichte des Lineals. Die Steifigkeit kann aus den ersten Untersuchungen im Rahmen unseres Experimentes abgeleitet werden (welche Kraft ergibt welche Verformung am Ende? Wie ergibt sich daraus die Steifigkeit?). Und danach braucht man sich nur noch zu erinnern an:

$$w = \sqrt{\frac{c}{m}}$$

und

$$f = \frac{w}{2p}$$

7.2 Variation der Eingabedaten

Am Anfang dieses Beispiels war bei der Idealisierung bei den geometrischen Daten, den Materialdaten und anderen Daten beurteilt worden, das sie mehr oder weniger Einfluss auf das Ergebnis - hier die Eigenfrequenzen - haben. Das sollten Sie prüfen.

Verändern Sie diese Daten etwas und vergleichen Sie die Simulationsergebnisse. Dies ist eine Variation, die besonders einfach mit der Simulation durchzuführen ist, denn

- wie einfach kann ein Zahlenwert der Eingabedaten geändert werden und die Berechnung wiederholt werden, aber
- wie mühsam ist es, ein anderes Lineal mit anderen Abmessungen oder anderen Werkstoffen zu besorgen!

Bei solchen "was wäre, wenn.." Aufgaben ist die Simulations-Methode unschlagbar vorteilhaft.

7.3 Variation des Elementtyps

Bei diesem Beispiel werden diejenigen, die in der Theorie der Finiten Elemente zuhause sind, stutzig werden. Es ist eine Anwendung, bei der die Biegung des Bauteils maßgebend ist. Hier wird eine Vernetzung verwendet, bei der über die Dicke des Bauteils ein einziges Element angeordnet ist. Dieses Element muss also an der Oberseite eine Zugspannung und an der Unterseite eine Druckspannung (bei der Schwingung entsprechend abwechselnd) abbilden. Der Theoretiker weiß, dass die üblichen Ansatzfunktionen für 3-dimensionale Elemente ohne Kantenmittenknoten eine solche Spannungsverteilung nicht abbilden

können. Warum ergeben sich hier trotzdem praxisgerechte Ergebnisse?

In diesem Beispiel ist der Elementtyp SOLID45 des ANSYS-Programms verwendet worden. Dieser Elementtyp ist in der Standardeinstellung (default) mit zusätzlichen Ansatzfunktionen ausgestattet, die für eine Aufgabe wie das Lineal sehr gut geeignet sind. Diese Ansatzfunktionen werden "extra displacement shapes" genannt. In der Literatur findet sich hierfür auch die Bezeichnung "enhanced strain". Mit diesen Erweiterungen ist dieser Elementtyp gut geeignet, auch mit einem Element über die Dicke eine Biegung der Struktur abzubilden. Dies trifft genau für unser Beispiel zu. Theoretisch hat diese Vorgehensweise den Nachteil, dass interne Inkompatibilitäten auftreten. Im Alltag sind diese Nachteile von untergeordneter Bedeutung, so dass für den Elementtyp die Standardeinstellung auf "include extra displacement shapes" eingestellt wurde.

Bei dem Programm ANSYS/Workbench wird der Elementtyp SOLID185 ebenfalls mit diesen Erweiterungen der Ansatzfunktionen "enhanced strain" verwendet. Daher kann auch dort eine Vernetzung mit einem Element über die Bauteildicke ein praxisgerechtes Ergebnis erreicht werden.

Um das Beispiel mit Elementen mit den üblichen Ansatzfunktionen für 3-dimensionale Elemente ohne Kantenmittenknoten zu berechnen, können Sie folgende Variante untersuchen:

Verwenden Sie das ANSYS-APDL-Programm mit dem Elementtyp SOLID45 und stellen Sie dabei die Keyoption (1) auf den Wert 1. Damit werden die zusätzlichen Ansatzfunktionen unterdrückt ("suppress extra displacement shapes"). Die Ergebnisse hiermit weichen sehr deutlich von den vorher berechneten Werten ab.

Die Standardeinstellungen sind so gewählt worden, dass im Ingenieuralltag automatisch eine praxisgerechte Vorgehensweise bei der Vernetzung und bei der Berechnung verwendet wird. Sie können damit guten Gewissens Ihrem Auftraggeber ein Ergebnis wie hier in diesem Beispiel abliefern. Wenn Sie allerdings eine theoretische Diskussion über die Ansatzfunktionen und die internen Inkompatibilitäten führen müssen, sollten Sie auch über die Alternativen der Vernetzung und der Einstellungen zu den Ansatzfunktionen informiert sein.

Das Beispiel hier stellt mit der einfachen Geometrie einen Ausnahmefall dar. In der Praxis sind meistens komplexere Geometrien und Querschnitte zu modellieren und zu berechnen. In diesen Fällen wird auch die Netzteilung entsprechend feiner sein. Und sind diese Fragen von sehr viel geringerer Bedeutung.

Dieses Beispiel hier kann auch gut mit Schalenelementen modelliert und berechnet werden. Eine solche Variante wird auch zum Selbststudium empfohlen. Schalenelemente sind flächig ausgedehnt. Die Abmessung in der dritten Dimension wird als ein zusätzlicher Zahlenwert übergeben (real constant), der nicht als Knoten-Koordinaten verarbeitet wird, sondern als Element-Eigenschaft.